
Bomen met Pythagoras
Misschien weet je al wat een Pythagorasboom is. Dat is een fractal. 
Een eindeloze herhaling van een simpel patroon. In dit geval begin 
je met een simpele streep, zoals hiernaast. Dat zal straks de stam 
van de Pythagorasboom worden.

Bovenop de stam komt de eerste vertakking: één tak gaat met 45° naar 
links, de andere met 45° naar rechts. En natuurlijk maak je ze iets 
kleiner, omdat takken dunner zijn dan de stam. Je krijgt dan het 
linkerfiguur.

   Je snapt wel dat je dit patroon 
kunt herhalen. Bovenop de eerste 
takken komen dan weer nieuwe 
vertakkingen. Zo krijg je het 
figuur aan de rechterkant.

Als je dit steeds maar blijft herhalen, krijg je 
een figuur dat we de Pythagorasboom noemen.

Een derde dimensie

Het figuur op de vorige pagina 
heeft natuurlijk wel iets weg van 
een boom, maar het ziet er toch 
maar kunstmatig uit.

Je kunt het figuur een beetje 
draaien om de stam (zoals 
hiernaast), maar het blijft 
hartstikke plat. Een saaie 
tweedimensionale boom in drie 
dimensies.

Om de boom echt driedimensionaal te maken, moeten we de stappen een 
klein beetje aanpassen. Nu is elke vertakking nog in het platte vlak. 
Maar niets weerhoudt ons ervan in elke stap de twee nieuwe takken een 
derde dimensie te geven.

   Hiernaast zie je de eerste tweede stappen van 
onze Pythagorasboom. Op elke tak zetten we 
opnieuw twee twijgen, ieder onder een hoek van 
45°. Maar ditmaal draaien we de tak zelf ook nog 
eens 90° om zijn as!

En dat helpt! In de 
derde stap zie je nog duidelijker wat het 
effect is. Elk paar takken staat 90° 
gedraaid ten opzichte van hun twijgen.

Als je dit oneindig blijft herhalen, krijg 
je een driedimensionale Pythagorasboom 
die er plotseling heel natuurlijk uit ziet.

Een goed voorbeeld van hoe fractals gebruikt kunnen worden om 
natuurlijke afbeeldingen te maken.

Zelf maken

De plaatjes zijn gemaakt met POV-ray. Dat is een open source (gratis) 
raytracingprogramma. Je kunt eenvoudig programmeren wat je ziet en 
hoe je het ziet. POV-ray rekent uit hoe het beeld (inclusief schaduwen, 
enz.) eruit komt te zien.

Op de volgende pagina staat het bestand dat gebruikt is om de laatste 
boom te genereren. Probeer vooral ook zelf de hoeken van de 
vertakkingen aan te passen of de verkleiningsfactor per stap. Misschien 
kun je wel een derde tak per stap toevoegen.

Download POV-ray op http://www.povray.org/download/ 

Martijn Leisink @ www.wiskunstelaar.nl

// plaats de camera, lichtbron en een witte achtergrond
camera { location <0,2,-3.5> look_at <0,1.6,0> }
light_source { <1000,2000,-3000> rgb 1 }
background { rgb <1,1,1> }

// een macro die zichzelf aanroept om de boom te maken
#macro pythagorasboom(diepte)
union {

// de diepte bepaalt hoelang er doorvertakt wordt
#if(diepte > 0)

object { // de eerste vertakking onder -45°
pythagorasboom(diepte-1)
scale 1/sqrt(2) // volgende tak iets kleiner 
rotate 90*y
rotate -45*z
translate <0,1,0> // en bovenop de vorige

}
object { // de tweede vertakking onder 45°

pythagorasboom(diepte-1)
scale 1/sqrt(2)  
rotate 90*y
rotate 45*z
translate <0,1,0>

}
#end
cone { // dit is een tak; die loopt wat taps toe

<0,0,0>, 0.1
<0,1,0>, 0.1/sqrt(2)

}        
sphere { // een bol ter hoogte van de vertakking

<0,1,0>, 0.1/sqrt(2)
}  

} 
#end

object {
pythagorasboom(15)
rotate -30*y
pigment { rgb <0,0.6,0> } // groen kleurtje

}

